


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L.1000

PER IL TUO **Commodore**

Una pubblicazione della J.soft editrice

sped. in abb. post. Gruppo II/70



PAPER

J.soft

35

Anno 2 - N° 35 - 13 settembre 1985

VIC=20

**Quattro
carte**

VIC=20

Il pescatore

**Giorgio il
giardiniere**

CI6=

STM

CI6=

C64=

**Dieta equilibrata
Keypad**

C64=

**Indovina i
colori**

Super drums

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono 'parole' racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [()] deve essere premuto con-

temporaneamente al tasto COMMODORE. Inoltre, se tra parentesi graffa si trova un carattere alfabetico "solitario" questo dovrà essere premuto contemporaneamente al tasto CTRL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori

{CLR}	SHIFT CLR HOME		{CYN}	CTRL 4		[<7>]		
{HOME}	CLR HOME		{PUR}	CTRL 5		[<8>]		
{SU}	SHIFT CLR SP		{GRN}	CTRL 6		{F1}		
{GTU}	CLR SP		{BLU}	CTRL 7		{F2}		
{SIN}	SHIFT CLR SP		{YEL}	CTRL 8		{F3}		
{DES}	CLR SP		[<1>]			{F4}		
{RVS}	CTRL 9		[<2>]			{F5}		
{OFF}	CTRL 0		[<3>]			{F6}		
{BLK}	CTRL 1		[<4>]			{F7}		
{WHT}	CTRL 2		[<5>]			{F8}		
{RED}	CTRL 3		[<6>]					

Per ricevere gli arretrati di **PAPER** compilare il seguente tagliando:



Compilare ed inviare in busta chiusa a:
J.soft viale Restelli, 5 - 20124 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di

PAPER _____ Anno _____

PAPER dedicato a _____ Anno _____

a L. 2.000 cad. per un totale di L. a mezzo

☐ assegno allegato

☐ contanti allegati





VIC-20

- 6** Quattro carte
di P. Shelk trad. e adatt. di M. Anticoli
- 9** Il pescatore
di P. Braddock trad. e adatt. di M. Anticoli

C64

- 16** Dieta equilibrata (1 parte) (stampante opz.)
di G. Valentini
- 23** Keypad
di D. Santinoli
- 25** Indovina i colori
di R. Michelotti
- 29** Super drums
di J. Ryan trad. e adatt. di M. Anticoli

CI64-plus/4

- 11** Gioglio il giardiniere (joystick)
di K. Woram trad. e adatt. di F. Sarcina
- 15** STM
di P. Gravelor trad. e adatt. di F. Sarcina

J.soft s.r.l.
**DIREZIONE, REDAZIONE
E AMMINISTRAZIONE**
Via Restelli, 5
20124 Milano
Tel.: 68.68.228-68.37.97

DIRETTORE RESPONSABILE:
Pietro Dell'Orco

COORDINAMENTO TECNICO:
Mauro Cristulib Grizzi

REDAZIONE:
Franco Sarcina

GRAFICA E IMPAGNAZIONE
Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOMPOSIZIONE:
Graphotek
Via Astesani, 16 - Milano
Tel. 64.80.397

CONTABILITÀ:
Giulia Pedrazzini
Flavia Bonatti

**AUTORIZZAZIONE ALLA
PUBBLICITÀ:**
Tribunale di Milano n° 178
del 14.04.1984

STAMPA:
Intergrafica - Pioltello (MI)
Rivista associata
all'Unione stampa
Periodica Italiana



PUBBLICITÀ
Concessionario per l'Italia e l'Estero
J.Advertising s.r.l.
V.le Restelli, 5
20124 MILANO
Tel. (02)
68.82.895-68.80.606-68.87.233
Tlx 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO

Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000
Numero arretrati L. 2.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Stop Bug

Se non l'hai già fatto, memorizza e salva su nastro o disco la versione di Stop Bug per il tuo computer. Questo programma ti permette, una volta caricato ed attivato con Run, di digitare i listati di Papersoft Commodore senza introdurre errori di battitura:

Come avrai notato, le linee dei nostri programmi terminano con la parola ":rem" seguita da un numero, detto checksum; non copiare questa parte della linea, che serve solo per tua informazione. Quando Stop Bug è in funzione, infatti, tutte le volte che premi il tasto Return dopo aver digitato una linea di programma, appare nell'angolo superiore sinistro dello schermo un numero, che deve coincidere con il checksum da noi pubblicato (sempre che tu non abbia usato

comandi abbreviati, come ? per Print).

Usando Stop Bug tieni conto del fatto che gli spazi non influiscono sui checksum, per cui fai particolare attenzione alle istruzioni Print dove sono spesso presenti. Inoltre eventuali inversioni di caratteri non vengono segnalate.

Un altro metodo per utilizzare Stop Bug, soprattutto se sei abituato ad usare le abbreviazioni dei comandi Basic, è controllare i checksum listando il programma e battendo Return su ogni linea.

Se intendi sospendere il lavoro di digitazione, puoi salvare su nastro o disco il programma che stai memorizzando senza che Stop Bug interferisca; al momento di riprendere il lavoro ricordati però di caricare ed attivare Stop Bug prima di ricaricare il tuo programma.

Per disabilitare Stop Bug premi Run/Stop e Restore (su C64 e VIC 20) oppure Run/Stop e Reset (su C16 e Plus/4). Per riattivarlo batti:

*SYS PEEK(56)*256 per il VIC 20;
SYS 16128 per il C16 e il PLUS/4;
SYS 40704 per il C 64.*

STOP BUG per C64 e VIC 20

```
100 POKE55,0:POKE56,PEEK(56)-1:R=PEEK(56)*256:PRINT"{CLR}
    ATTENDERE PREGO..."
110 FORI=RTOR+132:READA:CK=CK+A:IFA=999THENA=R/256
120 POKEI,A:NEXT
130 IFCK<>22689THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
    TA":END
140 SYS(R):PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIVAT
    O.":NEW
200 DATA173,36,3,201,32,208,1,96,141
210 DATA33,999,173,37,3,141,34,999,169
220 DATA32,141,36,3,169,999,141,37,3
230 DATA169,0,133,254,96,32,87,241,133
240 DATA251,134,252,132,253,8,201,13,240
250 DATA17,201,32,240,5,24,101,254,133
260 DATA254,165,251,166,252,164,253,40,96
270 DATA169,13,32,210,255,165,214,141,251
280 DATA3,206,251,3,169,0,133,216,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,254,169,0
310 DATA133,254,172,33,999,192,87,208,6
320 DATA32,205,189,76,117,999,32,205,221
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,214,76,55,999
```

STOP BUG per C16 e PLUS/4

```
100 POKE55,0:POKE56,63:PRINT"{CLR}ATTENDERE PREGO..."
110 FORI=16128TO16260:READA:CK=CK+A:POKEI,A:NEXT
120 IFCK<>14248THIENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
    TA":END
130 SYS16128:PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIV
    ATO.":NEW
200 DATA173,34,3,201,32,208,1,96,141
210 DATA33,63,173,35,3,141,34,63,169
220 DATA32,141,34,3,169,63,141,35,3
230 DATA169,0,133,6,96,32,232,235,133
240 DATA3,134,4,132,5,8,201,13,240
250 DATA17,201,32,240,5,24,101,6,133
260 DATA6,165,3,166,4,164,5,40,96
270 DATA169,13,32,210,255,165,205,141,251
280 DATA3,206,251,3,169,0,133,207,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,6,169,0
310 DATA133,6,172,33,63,192,232,208,6
320 DATA32,95,164,76,117,63,32,7,208
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,205,76,55,63
```

Quattro carte

Quello che vi presentiamo è un semplice gioco d'azzardo, nel quale ad ogni mano vengono scoperte quattro carte, contraddistinte unicamente dal loro seme. All'inizio disponete di 25 gettoni, e ad ogni mano ne dovete mettere in gioco uno. Il computer, che fa da croupier, paga tre differenti combinazioni: con quattro carte

diverse si vincono sette gettoni, con quattro carte uguali se ne vincono ancora sette, e infine con tre carte uguali la vincita è di tre gettoni. Il gioco termina se arrivate ad accumulare almeno 50 gettoni, sbancando in computer, oppure se perdete tutto.

Ed ora... buona fortuna!

```

10 REM  QUATTRO CARTE                                :rem  228
20 GOSUB960                                           :rem  129
30 GOSUB80                                           :rem   75
40 GOSUB150                                          :rem  122
50 GOSUB230                                          :rem  122
60 GOSUB490                                          :rem  131
70 GOTO660                                           :rem   60
80 PRINT"{CLR}":PRINT" POSSIEDI ";BB;"GETTONI":PRINT:PRIN :rem   29
   T                                                  :rem  175
90 GOSUB1140
100 PRINT"{ 4 SPAZI}{RVS}PREMI UN TASTO","{OFF}
    { 4 SPAZI}{RVS} PER{ 3 SPAZI}GIRARE "          :rem  212
110 PRINT"{OFF}{ 7 SPAZI}{RVS}LE CARTE"            :rem    7
120 GETA$                                           :rem  216
130 IFA$=""THEN120                                  :rem  203
140 RETURN                                           :rem  117
150 C=0:H=0                                          :rem   53
160 S=0:D=0                                          :rem   66
170 PRINT"{HOME}{RVS}{BLK}{ 12 GIU'}LE CARTE GIRATE SONO :rem   62
    :{BLU}"
180 PRINT"{RVS}{ 2 DES}{ 3 SPAZI}{ 2 DES}{ 3 SPAZI}
    { 2 DES}{ 3 SPAZI}{ 2 DES}{ 3 SPAZI}"          :rem  100
190 A=A+1                                           :rem  181
200 IFA=5THEN220                                     :rem  151
210 GOTO180                                          :rem  101
220 A=0:RETURN                                       :rem   92
230 ONINT(RND(1)*4)GOTO260,280,300                 :rem  136
240 H=H+1:J=2                                       :rem  178
250 I=83:GOTO320                                    :rem  144
260 D=D+1:J=2                                       :rem  172
270 I=90:GOTO320                                    :rem  144
280 S=S+1:J=0                                       :rem  202

```

290 I=65:GOTO320	:rem	148
300 C=C+1:J=0	:rem	163
310 I=88	:rem	138
320 K=K+5	:rem	200
330 POKE8030+K,I	:rem	123
340 POKE38750+K,J	:rem	185
350 L=L+1	:rem	201
360 GOSUB410	:rem	174
370 IFL=4THEN400	:rem	169
380 FORM=1TO500:NEXTM	:rem	59
390 GOTO230	:rem	106
400 L=0:K=0:RETURN	:rem	89
410 FORNS=1TOYY	:rem	224
420 POKEV,15	:rem	173
430 POKEF,250	:rem	207
440 FORY=1TO10:NEXTY	:rem	28
450 POKEV,0	:rem	122
460 POKEF,0	:rem	107
470 NEXTNS	:rem	123
480 YY=0:RETURN	:rem	213
490 GOSUB910	:rem	183
500 IFH=1ANDD=1ANDS=1ANDC=1THENYY=12:GOTO560	:rem	6
510 IFH=4ORD=4ORS=4ORC=4THENYY=12:GOTO600	:rem	120
520 IFH=3ORD=3ORS=3ORC=3THENYY=5:GOTO630	:rem	74
530 PRINT"{HOME}":PRINTTAB(6)"HAI{ 2 SPAZI}PERSO"	:rem	64
540 BB=BB-1:GOSUB1140	:rem	188
550 RETURN	:rem	122
560 PRINT"{HOME}QUATTRO CARTE DIVERSE!"	:rem	81
570 GOSUB410	:rem	177
580 BB=BB+7:GOSUB1140	:rem	196
590 RETURN	:rem	126
600 PRINT"{HOME}QUATTRO CARTE UGUALI !"	:rem	1
610 BB=BB+7:GOSUB410	:rem	141
620 GOSUB1140:RETURN	:rem	248
630 PRINT"{HOME}{ 2 SPAZI}TRE CARTE UGUALI !"	:rem	191
640 BB=BB+3:GOSUB410	:rem	140
650 GOSUB1140:RETURN	:rem	251
660 IFBB<1THEN690	:rem	234
670 IFBB>49THEN790	:rem	42
680 GOTO30	:rem	58
690 POKE36879,8	:rem	67
700 PRINT"{CLR}"	:rem	251
710 PRINT"{RVS}{WHT}SEI IN BANCAROTTA !!"	:rem	25
720 POKEV,10	:rem	171
730 FORQ=250TO150STEP-4	:rem	124
740 POKEF,Q	:rem	141
750 FORM=1TO100	:rem	114
760 NEXTM	:rem	41
770 NEXTQ	:rem	46
780 POKEV,0:POKEF,0:END	:rem	156

790 POKEV,15	:rem	183
800 FORQ=150TO250STEP4	:rem	77
810 POKEF,Q	:rem	139
820 FORM=1TO100	:rem	112
830 NEXTM	:rem	39
840 NEXTQ	:rem	44
850 POKE36879,110	:rem	155
860 PRINT"{CLR}"	:rem	2
870 PRINT"{ 2 SPAZI}{RVS}{WHT}MI HAI SBANCATO !!"	:rem	124
880 W=W+1	:rem	231
890 IFW=22THENPOKEV,0:POKEF,0:END	:rem	84
900 GOTO870	:rem	113
910 POKE7680+O,32	:rem	169
920 O=O+1	:rem	210
930 IFO=88THEN950	:rem	244
940 GOTO910	:rem	112
950 O=0:RETURN	:rem	116
960 POKE36879,28	:rem	117
970 PRINT"{CLR}{BLU}{RVS}***LE QUATTRO CARTE***"	:rem	97
980 A=INT(RND(1)*2)+1	:rem	121
990 IFA=1THENB=0	:rem	190
1000 IFA=2THENB=2	:rem	224
1010 ONINT(RND(1)*4)GOTO1030,1040,1050	:rem	63
1020 C=65:GOTO1060	:rem	232
1030 C=83:GOTO1060	:rem	233
1040 C=88:GOTO1060	:rem	239
1050 C=90:GOTO1060	:rem	233
1060 FORE=1TO2:NEXT	:rem	175
1070 POKE7702+D,C	:rem	165
1080 POKE38422+D,B	:rem	216
1090 D=D+1	:rem	235
1100 IFD=66THEN1120	:rem	49
1110 GOTO980	:rem	157
1120 A=0:B=0:C=0:D=0:V=36878:F=36876	:rem	216
1130 BB=25	:rem	237
1140 FORZ=1TO1000:NEXTZ	:rem	172
1150 RETURN	:rem	167

Il pescatore

A chi non piace in questa stagione fare un bel bagno nel mare, magari praticando la pesca subacquea? In questo gioco sarai appunto un pescatore subacqueo, che deve cercare di catturare il maggior numero possibile di pesci evitando di finire in pasto agli squali, che abbondano nel tratto di mare dove si svolge la pesca. Per muovere il nostro simpatico protagonista usa i seguenti tasti:

Q per andare verso l'alto

Z per andare verso il basso

I per andare verso sinistra

P per andare verso destra

Hai a disposizione una bombola da 300 litri di ossigeno; la partita termina se vieni a contatto con uno squalo oppure resti senza ossigeno. Per avere una bombola nuova, comunque, è sufficiente riemergere. Alla fine del gioco il computer esprimerà un giudizio sulla tua abilità di subacqueo.

```

10 REM *IL PESCATORE*                                :rem 212
20 REM ** PAPERSOFT **                                :rem 162
30 X=16:Y=5:M=4:POKE650,255:PRINTCHR$(8):POKE56,28    :rem 227
40 POKE36879,238:PRINT"{CLR}{BLK}{ 2 GIU'}{ 2 SPAZI}ATTEN :rem 173
   DERE PREGO...":GOSUB120
50 FORI=7834TO8163:POKEI+30720,INT(RND(1)*8):NEXT      :rem 159
60 FORT=1TO40:POKE7834+RND(1)*330,(RND(1)*2)+2:NEXT   :rem 187
70 FORT=1TO20:GOSUB380:NEXT                             :rem 227
80 FORI=7812TO7825:POKEI+30720,6:POKEI,192:NEXT      :rem 78
90 FORI=7826TO7833:POKEI+30720,2:POKEI,248:NEXT      :rem 81
100 FORI=8164TO8185:POKEI+30720,7:POKEI,249:NEXT:rem 124
110 GOTO230                                              :rem 96
120 FORP=0TO46:READA:POKE828+P,A:NEXT:FORP=0TO71:READA:PO :rem 103
   KE7168+P,A:NEXT
130 FORP=7424TO7432:POKEP,0:NEXT:POKE36869,255:PRINT" :rem 211
   {CLR}":RETURN
140 DATA169,154,162,16,133,87,169,30,133             :rem 233
150 DATA88,160,0,177,87,133,89,200,177               :rem 142
160 DATA87,136,145,87,200,152,201,21,208            :rem 221
170 DATA244,165,89,145,87,165,87,24,105             :rem 201
180 DATA22,133,87,144,2,230,88,202,208,220,96       :rem 214
190 DATA0,0,4,140,252,140,4,0,1,3,7,127,255,3,30,0  :rem 154
200 DATA0,0,0,0,52,124,52,0,0,57,127,57,0,0,0,0    :rem 247
210 DATA58,58,18,254,58,58,40,108,108,40,58,58,254,18,58, :rem 169
   58,8,136,251,63,251,136,63,0

```

```

220 DATA 16,17,223,252,223,17,252,0,8,28,8,0,12,104,105,12
    7                                     :rem 59
230 P=32:P2=7:U=300:GOSUB320             :rem 101
240 FOR T=1 TO 200:NEXT:POKE7680+X+22*Y,P:POKE38400+X+22*Y,P
    2                                     :rem 231
250 SYS828:GOSUB320:U=U-1               :rem 6
260 GETA$:IFA$="Q" THEN Y=Y-1:M=4:P2=7:IFY<5 THEN Y=5
                                           :rem 14
270 IFY<7 THEN U=300                     :rem 72
280 IFA$="Z" THEN Y=Y+1:M=5:P2=7:IFY>23 THEN Y=23 :rem 251
290 IFA$="P" THEN X=X+1:M=6:P2=7:IFX>21 THEN X=21 :rem 235
300 IFA$="I" THEN X=X-1:M=7:P2=7:IFX<0 THEN X=0   :rem 119
310 GOSUB560:GOTO240                     :rem 184
320 PRINT "{HOME}{BLK}{RVS}PESCI: ";F         :rem 74
330 PRINT "{GIU'}{ 19 SPAZI}{SU}":IFU<0 THEN GOTO390
                                           :rem 151
340 PRINT "{RVS}ARIA : ";U:RETURN          :rem 123
350 D=INT(RND(1)*330)+7834                :rem 124
360 IFPEEK(D)<>32 THEN 350                 :rem 136
370 POKED,INT(RND(1)*2)+2:RETURN           :rem 174
380 R=RND(1)*350:POKE7834+R,1:POKE7834+R+1,0:RETURN
                                           :rem 249
390 POKE7680+X+22*Y,8:POKE38400+X+22*Y,0:PRINT "{HOME}
    {RVS}{ 12 GIU'}{ 6 DES}SEI MORTO!{OFF}"      :rem 193
400 FOR I=1 TO 1000:NEXT:POKE7680+X+22*Y,32     :rem 115
410 FOR T=Y TO 0 STEP -1:P=PEEK(7680+X+22*T):P2=PEEK(38400+X+2
    2*T)                                           :rem 115
420 POKE38400+X+22*T,1:POKE7680+X+22*T,8       :rem 227
430 FOR I=1 TO 500:NEXT                     :rem 230
440 POKE7680+X+22*T,P:POKE38400+X+22*T,P2      :rem 78
450 NEXT T:PRINT "{CLR}{BLU}":POKE36879,27:POKE36869,240
                                           :rem 51
460 IFF>=100 THEN A$="SUPERBO !!"           :rem 242
470 IFF<100 THEN A$="BUONO"                :rem 213
480 IFF<70 THEN A$="PASSABILE"             :rem 189
490 IFF<40 THEN A$="FANNULLONE !!"         :rem 91
500 IFF<20 THEN A$="POCO DI BUONO !!"      :rem 160
510 FOR TD=1 TO 1000:NEXT:PRINT "{CLR}{ 2 GIU'}{BLU}IL TUO GR
    ADO E':{GIU'}":PRINT A$                     :rem 28
520 PRINT "{ 3 GIU'}{BLK} GIOCHI ANCORA? (S/N) " :rem 242
530 GETA$:IFA$<>"S" AND A$<>"N" THEN 530      :rem 39
540 IFA$="S" THEN RUN                      :rem 133
550 PRINT "{CLR}{BLU}":END                 :rem 46
560 POKE198,0:P=PEEK(7680+X+22*Y):P2=PEEK(38400+X+22*Y)
                                           :rem 114
570 IFP=0 OR P=1 THEN 390                  :rem 18
580 IFP<4 THEN F=F+1:P=32:GOSUB350         :rem 190
590 POKE7680+X+22*Y,M:POKE38400+X+22*Y,7:RETURN :rem 42

```


Giorgio il giardiniere

joystick

Giorgio possiede un bellissimo giardino dove coltiva fiori stupendi. Deve però difenderlo costantemente dall'attacco delle erbacce e dei parassiti, spruzzando l'erbicida sulle prime e l'antiparassitario sui fiori.

In questo gioco sarete voi a guidare Giorgio nella sua opera di disinfestazione (per mezzo del joystick n. 1), facendogli irrorare tutti i fiori ed evitando nel contempo che le erbacce si diffondano troppo (usate il pulsante per spruzzare i liquidi). In tal modo totalizzerete punti, ma dovrete lottare contro il tempo, in quanto avete solo due minuti a disposizione per spruzzare tutti i fiori. Si ottengono 10 punti per le margherite gialle, 20 per quelle blu e 40 per le rose; una

volta irrorati, i fiori divengono bianchi. Inoltre ogni erbaccia eliminata frutta 5 punti ma, ripetiamo, non è necessario ucciderle tutte. Come abbiamo detto, l'opera è completa quando tutti i fiori sono stati irrorati, ma il gioco viene interrotto se vi sono troppe erbacce: per poter evitare che ciò si verifichi è utile sapere che esse appaiono dapprima come semi, poi come germogli ed infine raggiungono lo stato adulto nel quale si riproducono generando nuovi semi. Se completerete il lavoro entro il tempo a vostra disposizione, vi troverete di fronte ad un nuovo giardino con più fiori e più erbacce, ma per ripulirlo disporrete di 15 secondi in più rispetto al precedente.

```

100 COLOR0,1:COLOR4,1:VOL8:GOTO390           :rem    89
110 CO=-1024                                   :rem    85
120 GOSUB1130:DIMP(200),JP(15),CS(15),DN$(15) :rem   254
130 SC=0:D=15:E=3:QW=E:ET=200:WL=20:LV=1:DN$="{HOME}"
    { 12 GIU' }"                               :rem   195
140 RS=439:AV=3152:U=40:B=32:FC=71:FI=0:GOSUB570:TI$="000
    000"                                         :rem   206
150 PRINT"{CLR}{WHT}PUNTI:"SC:PRINT"{HOME}"TAB(14)"LIVELL
    O:"LV:PRINT"{HOME}"TAB(27)"TEMPO: "TI$;    :rem   239
160 POKE3112+CO,69:POKE3112,85:FORK=3113TO3150:POKEK+CO,6
    9:POKEK,67:NEXT                            :rem    93
170 POKE3151+CO,69:POKE3151,73                :rem   178
180 FORK=3152TO3992STEP40:POKEK+CO,69:POKEK,66:POKEK+CO+3
    9,69:POKEK+39,66:NEXT                      :rem    68
190 POKE4032+CO,69:POKE4032,74:FORK=4033TO4070:POKEK+CO,6
    9:POKEK,67:NEXT                            :rem   102
200 POKE4071+CO,69:POKE4071,75:PL=93:QF=0     :rem   111
210 FORK=1TO3:FORJ=1TOD                       :rem   201

```

```

220 FP=(INT(RND(1)*RS)*2)+AV:SP=FP+U           :rem 119
230 IFPEEK(FP)<>BTHEN220                         :rem 175
240 IFPEEK(SP)<>BTHEN220                         :rem 189
250 POKEFP+CO,FC:POKEFP,PL:POKESP+CO,69:POKESP,109:QF=QF+
    1:NEXT:D=D-10                               :rem 210
260 IFFC=71THENFC=67:PL=94:GOTO280             :rem 68
270 FC=66:PL=95                               :rem 76
280 NEXT:D=D+30:OP=3153:POKEOP,98:WC=105:X=0:GOSUB340:TI$
    ="000000":GOTO330                         :rem 137
290 IFH<>0THENH=0                             :rem 0
300 IFE<0THENE=0                             :rem 180
310 IFWC=108THENWC=105:X=X+E:GOSUB340:GOTO330 :rem 32
320 GOSUB380                                   :rem 176
330 L=TI+500:GOTO620                          :rem 135
340 FORH=XTOX+E                               :rem 198
350 WP=INT(RND(1)*(RS*2))+AV:IFPEEK(WP)<>BTHEN350
                                                :rem 197
360 PP(H)=WP:POKEPP(H)+CO,73:POKEPP(H),WC:NEXT:E=E+1:IFH>
    WLTHEN1010                                :rem 204
370 RETURN                                     :rem 122
380 FORH=XTOX+E-1:POKEPP(H)+CO,73:POKEPP(H),WC:NEXT:RETUR
    N                                           :rem 24
390 :                                           :rem 214
400 GOSUB1330:BC=15080                       :rem 213
410 FORK=1TO4:FORM=BC TO BC+7:READCD:POKENM,CD:NEXT:RESTOR
    E:BC=BC+8:NEXT                             :rem 220
420 BC=BC-8:FORM=BC TO BC+111:READCD:POKENM,CD:NEXT:GOTO11
    0                                           :rem 184
430 DATA0,0,0,0,24,126,231,60               :rem 160
440 DATA24,60,66,153,189,255,126,60         :rem 244
450 DATA60,126,255,189,153,66,60,24         :rem 245
460 DATA28,38,79,223,223,79,38,28           :rem 160
470 DATA192,252,70,95,95,127,62,28          :rem 205
480 DATA28,62,127,95,95,70,252,192          :rem 206
490 DATA56,100,242,251,251,242,100,56        :rem 67
500 DATA3,63,98,250,250,254,124,56          :rem 187
510 DATA56,124,254,250,250,98,63,3          :rem 188
520 DATA0,0,24,24,24,0,0,0                 :rem 7
530 DATA0,34,28,8,28,34,0,0                 :rem 80
540 DATA66,231,126,60,36,126,231,66         :rem 236
550 DATA153,60,90,255,255,90,60,153         :rem 240
560 DATA24,24,153,219,126,60,24,0          :rem 127
570 :                                           :rem 214
580 JP(0)=0:JP(1)=-40:JP(2)=-39:JP(3)=1:JP(4)=41:rem 55
590 JP(5)=40:JP(6)=39:JP(7)=-1:JP(8)=-41    :rem 134
600 CS(0)=97:CS(1)=97:CS(2)=103:CS(3)=102:CS(4)=104
                                                :rem 200
610 CS(5)=98:CS(6)=101:CS(7)=99:CS(8)=100:RETURN:rem 207
620 JV=JOY(1)AND127:FR=JOY(1)AND128:CS(0)=CS(JV):rem 92
630 IFFR=128THENGOSUB690                     :rem 230
640 NP=OP+JP(JV):IFPEEK(NP)<>BTHENNP=OP       :rem 12

```



```

650 POKENP+CO,79:POKEOP,B:POKENP,CS(JV):OP=NP      :rem 165
660 IFTI>LTHENWC=WC+1:GOTO290                        :rem 92
670 PRINT" {HOME}"TAB(34)TI$;:IFVAL(TI$)>ETTHEN1020  :rem 147
680 GOTO620                                           :rem 111
690 MP=CS(JV):GOSUB810                               :rem 83
700 G=NP+CP:CM=G+CO:TP=PEEK(G):IFTP<90ANDTP>32THENRETURN :rem 149
710 POKECM,65:POKEG,108                             :rem 99
720 FORSN=1TO2:SOUND3,900,8                         :rem 131
730 FORH=1TO150:NEXT:NEXT                           :rem 98
740 IFTP=109THENPOKECM,69:POKEG,109:RETURN          :rem 190
750 IFTP=96THENPOKECM,113:POKEG,96:RETURN           :rem 143
760 IFTP=BTHENPOKEG,B:RETURN                         :rem 124
770 IFTP=93THENSC=SC+10:GOTO890                      :rem 178
780 IFTP=94THENSC=SC+20:GOTO890                      :rem 181
790 IFTP=95THENSC=SC+40:GOTO890                      :rem 185
800 SC=SC+5:E=E-1:POKEG,B:PRINT" {HOME} { 7 DES}";SC:RETURN :rem 178
810 IFMP=97THENCNP=-40:RETURN                        :rem 28
820 IFMP=98THENCNP=40:RETURN                         :rem 241
830 IFMP=99THENCNP=-1:RETURN                        :rem 237
840 IFMP=100THENCNP=-41:RETURN                      :rem 65
850 IFMP=101THENCNP=39:RETURN                       :rem 29
860 IFMP=102THENCNP=1:RETURN                        :rem 228
870 IFMP=103THENCNP=-39:RETURN                      :rem 78
880 CP=41:RETURN                                     :rem 239
890 POKECM,113:POKEG,96:PRINT" {HOME} { 7 DES}";SC:rem 38
900 FI=FI+1:IFFI=QFTHEN920                          :rem 70
910 RETURN                                           :rem 122
920 PRINTDN$TAB(10)"LIVELLO";LV;"COMPLETATO {GIU'}":TL=ET-VAL(TI$) :rem 249
930 PRINTTAB(15)"BONUS:";TL                         :rem 151
940 SC=SC+TL:LV=LV+1:D=D+4:QW=QW+1:E=QW:WL=WL+1:ET=ET+15 :rem 32
950 IFINT(ET/100+.5)>INT(ET/100)THENET=INT(ET/100+.5)*100 :rem 6
960 IFD>70THEND=70                                  :rem 46
970 IFET>500THENET=500                              :rem 53
980 IFQW>20THENQW=20                                :rem 238
990 IFWL>40THENWL=40                                :rem 233
1000 FORK=1TO2000:NEXT:GOTO140                      :rem 71
1010 PRINTDN$TAB(13)"TROPPE ERBACCE!":GOTO1030      :rem 236
1020 PRINTDN$TAB(13)"TEMPO SCADUTO!"               :rem 143
1030 IFSC>HSTHENHS=SC                               :rem 95
1040 FORJ=1TO2000:NEXT                             :rem 66
1050 PRINT" {CLR}"TAB(9)"MIGLIOR PUNTEGGIO:";HS    :rem 164
1060 PRINTTAB(9)" {GIU'}IL TUO PUNTEGGIO E'";SC    :rem 134
1070 PRINTTAB(7)" {GIU'}HAI RAGGIUNTO IL LIVELLO ";LV :rem 20

```

```

1080 PRINTDN$"{ 7 GIU' }"TAB(5)"{PUR}PREMI IL PULSANTE PER
      RIGIOCARE"                                     :rem 254
1090 PRINTTAB(11)"{GIU'}{RVS}RETURN{OFF} PER SMETTERE
      [<8>]"                                           :rem 165
1100 GETQ$:M=JOY(1)AND128:IFM=128THEN130               :rem 110
1110 IFQ$=CHR$(13)THENPOKE65298,PEEK(65298)OR4:POKE65299,
      208:SCNCLR:END                                   :rem 249
1120 GOTO1100                                           :rem 191
1130 :                                                  :rem 255
1140 PRINT"{CLR}"TAB(8)"[<1>]{RVS} GIORGIO IL GIARDINIERE
      {OFF}{ 2 GIU'}{WHT}"                             :rem 226
1150 PRINTTAB(7)"USA IL JOYSTICK (PORTA 1){GIU'}"
      :rem 44
1160 PRINT"GIORGIO ( ) DEVE RAGGIUNGERE E IRRORARE{GIU'}"
      :rem 10
1170 POKE3281+CO,79:POKE3281,97                        :rem 242
1180 PRINT"LE ERBACCE ( ).{ 2 SPAZI}IL NOSTRO GIARDINIERE
      {GIU'}"                                           :rem 62
1190 POKE3364+CO,73:POKE3364,107                      :rem 26
1200 PRINT"DEVE ANCHE DISINFESTARE I SUOI FIORI.
      { 2 GIU'}"                                         :rem 234
1210 PRINTTAB(7)"** TABELLA DEI PUNTEGGI **{GIU'}"
      :rem 215
1220 PRINTTAB(10)"ERBACCE"SPC(9)"5 PUNTI{GIU'}" :rem 142
1230 FORI=0TO2:POKE3651+I*2+CO,73:POKE3651+I*2,105+I:NEXT
      :rem 74
1240 PRINT"MARGHERITE GIALLE"SPC(9)"10 PUNTI{GIU'}"
      :rem 228
1250 POKE3693+CO,71:POKE3693,93:POKE3733+CO,69:POKE3733,1
      09                                                :rem 131
1260 PRINT"{ 3 SPAZI}MARGHERITE BLU"SPC(9)"20 PUNTI{GIU'}
      "                                                 :rem 28
1270 POKE3773+CO,67:POKE3773,93:POKE3813+CO,69:POKE3813,1
      09                                                :rem 134
1280 PRINTTAB(13)"ROSE"SPC(9)"40 PUNTI{ 2 GIU'}":rem 43
1290 POKE3853+CO,66:POKE3853,93:POKE3893+CO,69:POKE3893,1
      09                                                :rem 149
1300 PRINTTAB(5)"{PUR}PREMI IL PULSANTE PER INIZIARE[<8>]
      "                                                 :rem 233
1310 M=JOY(1)AND128:IFM=128THENRETURN                 :rem 46
1320 GOTO1310                                           :rem 196
1330 REM INIZIO CARATTERI=14336 ($3800)               :rem 154
1340 COLOR0,1:PRINT"{CLR}{BLK}{ 3 GIU'}MO":PRINT"
      { 5 GIU'}TD000 D6FF 3800":PRINT"{GIU'}X":PRINT"
      {GIU'}GO1370{HOME}"                             :rem 46
1350 POKE51,0:POKE55,0:POKE52,56:POKE56,56           :rem 187
1360 POKE239,4:FORI=1319TO1322:POKEI,13:NEXT:END:rem 163
1370 POKE65298,PEEK(65298)AND251:POKE65299,56:RETURN
      :rem 76

```


STM

STM è una abbreviazione inglese che indica la memoria breve, cioè la capacità di ricordare le cose avvenute poco tempo prima. Questo semplice programma vi dà la possibilità di migliorare la vostra memoria breve in maniera divertente. All'inizio vi viene chiesto se volete cimentarvi con le lettere dell'alfabeto, con le cifre da 0 a 9, oppure solamente con i due simboli 0 - 1. Dopo che avete fatto la vostra scelta, vi viene presentata per un tempo molto breve una sequenza

di numeri o di lettere, che dovete imprimervi bene nella mente. Dovrete poi digitare i simboli nello stesso ordine in cui vi sono stati presentati. Se la risposta sarà esatta, vi verrà mostrata una nuova sequenza più lunga della precedente; si comincia infatti con un solo carattere, e ogni volta la sequenza si allunga di un simbolo. Potrete anche gareggiare con i vostri amici: vincerà chi avrà ricordato la sequenza più lunga impiegando il minor tempo possibile.

10 REM **STM**	:rem 225
20 COLOR4,9,3:COLOR0,16,4:PRINT">{CLR}{" 14 DES}{FLASH ON}	:rem 85
{RVS}{BLK} * S T M *	:rem 253
30 PRINT"{ 2 GIU' }VUOI GIOCARE CON..."	:rem 5
40 PRINT"{GIU' }1: LETTERE ?"	:rem 194
50 PRINT"{GIU' }2: NUMERI ?"	:rem 159
60 PRINT"{GIU' }3: SOLO CON 0-1 ?"	:rem 224
70 PRINT"{GIU' }4: VUOI TERMINARE ?"	:rem 173
80 GETA\$:rem 171
90 ONVAL(A\$)GOTO110,120,130,420	:rem 50
100 GOTO80	:rem 164
110 A=26:B=65:GOTO140	:rem 159
120 A=10:B=48:GOTO140	:rem 105
130 A=2:B=48	:rem 160
140 INPUT"{GIU' }SEI PRONTO <RETURN> ??";K\$:rem 218
150 IFK\$=" "THEN150	:rem 250
160 C=1:TT=300	:rem 38
170 T=TI:TA=T	:rem 70
180 A\$="":PRINT	:rem 25
190 FORD=1TOC	:rem 196
200 B\$=CHR\$(B+RND(0)*A)	:rem 134
210 PRINTB\$:rem 44
220 A\$=A\$+B\$:rem 243
230 FORE=1TOTT:NEXT	:rem 63
240 PRINT"{SIN} ";	:rem 214
250 NEXT	

260 PRINT	:rem	37
270 FORW=1TO250:NEXT:PRINT" {CLR}"	:rem	150
280 REM**RISPOSTA**	:rem	155
290 POKE239,0:INPUT"RISPOSTA";AN\$:rem	53
300 IFAN\$=A\$THEN400	:rem	58
310 PRINT"{GIU' } {RVS} SBAGLIATO! {GIU' }"	:rem	80
320 PRINT"PUNTI:";C	:rem	174
330 PRINT"LA RISPOSTA ERA: ";A\$:rem	27
340 PRINT"TEMPO:";STR\$(INT((TA-T)/.6)/100);" SECONDI"		
	:rem	74
350 PRINT"{GIU' } GIOCHI ANCORA (S/N) ?"	:rem	65
360 GETA\$:rem	222
370 IFA\$="S" THEN140	:rem	38
380 IFA\$<>"N" THEN360	:rem	99
390 GOTO10	:rem	54
400 C=C+1:TA=TI	:rem	92
410 GOTO180	:rem	103
420 END	:rem	109

Dieta equilibrata

prima parte

C64K-plus/4

Stampante opz.

Uno degli argomenti "top" che si incontrano leggendo i rotocalchi o guardando la televisione, specialmente durante il periodo estivo, è quello della dieta: come possiamo riuscire a rimanere o a ritornare magri senza essere costantemente affamati?

Con questo programma Papersoft ti permette di analizzare giornalmente la tua dieta e di scoprire quindi se essa risulta equilibrata. Non devi fare altro che impostare il tipo e il quantitativo in grammi dei cibi che hai mangiato nelle 24 ore per ottenere un preciso responso dietologico.

Il programma è diviso, a causa della sua lunghezza, in 3 parti: digita la prima ed aspetta pazientemente

(io approfitterei dell'attesa per gustarmi qualche abbondante panzetto!) i prossimi due numeri di Papersoft Commodore.

"Dieta Equilibrata" è di facilissimo uso, data la sua struttura a "menu", e contiene già le istruzioni necessarie per il suo corretto uso. Se possiedi una qualsiasi stampante Commodore puoi ottenere i dati degli alimenti e le stime dietologiche anche su carta. La versione pubblicata gira su C64, tuttavia con una semplice modifica il programma si adatta perfettamente al Plus/4. Chi possiede questo computer deve digitare la linea 110 nel seguente modo:

110 COLOR 4,6,6:COLOR0,1:CL=

-1024:CO=78:SC=3072

Il programma non è invece adatto

al C16 a causa della notevole memoria occupata.

```
100 PRINT"{CLR}" :rem 245
110 POKE53280,5:POKE53281,0:CL=54272:CO=14:SC=1024 :rem 56
120 FORT=0TO10 :rem 63
130 FORI=SC+41*TTOSC+39+39*T :rem 138
140 POKEI,114:POKEI+960-80*T,113:POKEI+CL,CO:POKEI+CL+960 :rem 80
    -80*T,CO :rem 78
150 NEXT:NEXT :rem 55
160 PRINT"{HOME}{WHT}{ 12 GIU'}{ 11 DES}{RVS}DIETA :rem 65
    { 2 SPAZI}EQUILIBRATA" :rem 108
170 FORQ=0TO10 :rem 185
180 FORW=SC+41*QTOSC+960-39*QSTEP40 :rem 74
190 POKEW,107:POKEW+39-2*Q,115:POKEW+CL,CO:POKEW+CL+39-2* :rem 250
    Q,CO :rem 145
200 NEXT:NEXT :rem 40
210 PRINT"{HOME}{GIU'}{ 24 DES}BY G.VALENTINI" :rem 242
220 GOTO3100 :rem 40
230 REM:VALUTAZIONE CAL.TOT. :rem 5
240 Q1$="{RVS} DIETA NORMOCALORICA:CONSUMO DI CALORIE :rem 177
    {OFF}{ 2 SPAZI}{RVS}ENTRO LIMITI ACCETTABILI" :rem 7
250 Q2$="{RVS} DIETA IPOCALORICA:CARENZA DI CALORIE" :rem 119
260 Q3$="{RVS} DIETA IPERCALORICA:ECCESSO DI CALORIE :rem 115
    {OFF}" :rem 42
270 IFCT<=CGANDCT>=CG-CG/8THENPRINTQ1$ :rem 98
280 IFCT<CG-CG/8THENPRINTQ2$ :rem 150
290 IFCT>CGTHENPRINTQ3$:PRINT"{ 2 SPAZI}L'ECCEDENZA E' DI :rem 153
    "CT-CG" CALORIE" :rem 199
300 RETURN :rem 108
310 REM:STAMPA VALUTAZ.VITAMINE :rem 122
320 IFS1$=""THENPRINT:PRINT"{ 2 SPAZI}{RVS} ASSENZA DI VI :rem 132
    TAMINA A" :rem 96
330 IFS2$=""THENPRINT:PRINT"{ 2 SPAZI}{RVS} ASSENZA DI VI
    TAMINA B1"
340 IFS3$=""THENPRINT:PRINT"{ 2 SPAZI}{RVS} ASSENZA DI VI
    TAMINA B2"
350 IFS4$=""THENPRINT:PRINT"{ 2 SPAZI}{RVS} ASSENZA DI VI
    TAMINA PP"
360 IFS5$=""THENPRINT:PRINT"{ 2 SPAZI}{RVS} ASSENZA DI VI
    TAMINA C"
370 RETURN
380 B#=0:OPEN3,0,1:PRINT"{GIU'}{ 2 DES}BATTI UN NUMERO PE
    R SCEGLIERE"
390 PRINT"{GIU'}{ 3 DES}QUALE VUOI{ 2 SPAZI}";:INPUT#3,B$
    :CLOSE3
```

```

400 IFASC(B$)<48ORASC(B$)>57THENB$="":B%=0:RETURN                                :rem 112
410 B%=VAL(B$):RETURN                                                            :rem 237
420 REM:STAMPA CALCOLO CARATTERISTICHE SINGOLO CIBO                                :rem 14
430 PRINT:PRINT:PRINT:PRINT"{CLR}"SPC(10)"{RVS}"A$(B%+I-1,1)                    :rem 138
440 PRINT:PRINT"CALORIE{ 13 C}"SPC(3)A$(B%+I-1,2)                                :rem 205
450 PRINT:PRINT"PROTEINE VEG.{ 7 C}"SPC(3)"GRAMMI"SPC(3)A$(B%+I-1,3)            :rem 31
460 PRINT:PRINT"PROTEINE ANIM.{ 6 C}"SPC(3)"GRAMMI"SPC(3)A$(B%+I-1,7)          :rem 164
470 PRINT:PRINT"GRASSI{ 14 C}"SPC(3)"GRAMMI"SPC(3)A$(B%+I-1,4)                :rem 202
480 PRINT:PRINT"CARBOIDRATI{ 9 C}"SPC(3)"GRAMMI"SPC(3)A$(B%+I-1,5)            :rem 88
490 PRINT:PRINT"VITAMINE{ 12 C}"SPC(3)A$(B%+I-1,6)                                :rem 113
500 PRINT:PRINT"{ 6 SPAZI}{ 18 S}"                                              :rem 3
510 PRINT:PRINT"{ 2 SPAZI}DATI{ 2 SPAZI}RIFERITI A 100 GR .DI ALIMENTO"         :rem 45
520 IFCA=1THEN610                                                                :rem 222
530 PRINT"{GIU'}QUANTO NE VUOI(IN GRAMMI)DI{GIU'}":PRINTTAB(10)A$(B%+I-1,1);:INPUTX :rem 100
540 PA=0:PV=0:G=0:Z=0:V$="":C=0                                                  :rem 242
550 PV=INT(VAL(A$(B%+I-1,3))*X/100*4):PA=INT(VAL(A$(B%+I-1,7))*X/100*4)         :rem 69
560 G=INT(VAL(A$(B%+I-1,4))*X/100*9):Z=INT(VAL(A$(B%+I-1,5))*X/100*4)          :rem 180
570 C=PV+G+Z+PA:V$=A$(B%+I-1,6)                                                :rem 183
580 IFX=0THENV$=""                                                              :rem 27
590 REM:STAMPA PROPRIETA' NUTRITIVE QUANTITA'SCELTA DI ALIMENTO                  :rem 89
600 PRINT"{CLR}"                                                                :rem 250
610 PRINT:PRINT"{ 2 SPAZI}DA"X"GRAMMI DI {RVS}"A$(B%+I-1,1):PRINT:PRINT"{OFF}{ 2 SPAZI}SI OTTENGONO:" :rem 239
620 PRINT:PRINT"PV"CALORIE IN "TAB(18)"{ 4 C}> {RVS} PROTEINE VEG."              :rem 87
630 PRINT:PRINT"PA"CALORIE IN "TAB(18)"{ 4 C}> {RVS} PROTEINE ANIM."            :rem 134
640 PRINT:PRINT"G"CALORIE IN "TAB(18)"{ 4 C}> {RVS} GRASSI"                    :rem 77
650 PRINT:PRINT"Z"CALORIE IN "TAB(18)"{ 4 C}> {RVS} ZUCCHERI"                  :rem 245
660 PRINT:PRINT"C;TAB(11)" { 10 C}> {RVS} CALORIE TOTALI"                      :rem 58
670 PRINT:PRINT"{ 2 SPAZI}"V$;TAB(15)"{ 9 C}>{ 2 SPAZI}{RVS} VITAMINE"         :rem 68
680 PRINT:FORTR=0TO19:PRINT"{RVS}QZ";:NEXT:PRINT:rem 175

```

690	IFCA=1 THEN 840	:rem	235
700	REM:MEMORIA ADDITIVA CARATTERISTICHE ALIMENTI SCELTI	:rem	123
710	AT=AT+PA:REM:PROT.ANIM.	:rem	217
720	VT=VT+PV:REM:PROT.VEG.	:rem	214
730	GT=GT+G:REM:GRASSI	:rem	160
740	ZT=ZT+Z:REM:ZUCCHERI	:rem	110
750	CT=CT+C:REM:CALORIE TOT.	:rem	241
760	REM:MEMORIA ADDITIVA VITAMINE	:rem	104
770	IFMID\$(V\$,1,2)>"{ 2 SPAZI}" THEN S1\$=MID\$(V\$,1,2)	:rem	203
780	IFMID\$(V\$,3,3)>"{ 3 SPAZI}" THEN S2\$=MID\$(V\$,3,3)	:rem	211
790	IFMID\$(V\$,6,3)>"{ 3 SPAZI}" THEN S3\$=MID\$(V\$,6,3)	:rem	219
800	IFMID\$(V\$,9,3)>"{ 3 SPAZI}" THEN S4\$=MID\$(V\$,9,3)	:rem	218
810	IFMID\$(V\$,12,2)>"{ 2 SPAZI}" THEN S5\$=MID\$(V\$,12,2)	:rem	46
820	VT\$=S1\$+S2\$+S3\$+S4\$+S5\$:rem	163
830	IFCA\$="S" THEN CA\$="":CA=1:OPEN1,4:CMD1:GOTO420	:rem	111
840	IFCA\$="" THEN CA=0:CA\$="S":PRINT#1:CLOSE1.	:rem	38
850	X=0:REM AZZERAMENTO GRAMMI ALIMENTO	:rem	230
860	PRINT"{OFF}{GIU'} PER CONTINUARE PREMI UN TASTO"	:rem	156
870	GETQW\$:IFQW\$="" THEN 870	:rem	47
880	RETURN:REM:RITORNA AL MENU PRINCIPALE	:rem	160
890	REM:STAMPA SUBMENU PANE	:rem	200
900	PRINT"{CLR}":I=1	:rem	238
910	PRINTTAB(10)"{ 2 GIU'}{RVS}PANE E SIMILARI{ 2 GIU'}"	:rem	13
920	FORR=1 TO 16	:rem	76
930	PRINTTAB(6)RTAB(11)A\$(R,1)	:rem	200
940	NEXT	:rem	220
950	GOSUB380	:rem	185
960	IFB%>16 OR B%<1 THEN 900	:rem	119
970	GOSUB420	:rem	182
980	RETURN	:rem	129
990	REM:STAMPA SUBMENU PRIMI	:rem	38
1000	PRINT"{CLR}":I=145	:rem	127
1010	PRINTSPC(16)"{RVS}PRIMI{GIU'}"	:rem	213
1020	FORR=145 TO 160	:rem	13
1030	PRINTR-144TAB(4)"{RVS}"A\$(R,1)"{OFF}"TAB(19)R-144+16 "{RVS}"A\$(R+16,1)	:rem	223
1040	NEXT	:rem	4
1050	PRINT"{GIU'}{RVS}NB:{OFF}DATI RIFERITI A 100 GR.+CON DIMENTI"	:rem	68
1060	GOSUB380	:rem	226
1070	IFB%>32 OR B%<1 THEN 1000	:rem	198
1080	GOSUB420	:rem	223


```

1090 RETURN :rem 170
1100 REM:STAMPA SUBMENU CARNI :rem 50
1110 PRINT"{CLR}":I=17 :rem 79
1120 FORE=1TO13:PRINT"{ 2 SPAZI}Q";:NEXT:PRINT :rem 188
1130 PRINTTAB(17)"{ 2 GIU'}{RVS}CARNI{GIU'}" :rem 216
1140 PRINT" {RVS}FRATTAGLIE"TAB(23)"CARNI FRESCHE"
:rem 202
1150 PRINT" {RVS}& AFFETTATI{GIU'}" :rem 121
1160 FORR=17TO29 :rem 180
1170 PRINTR-17+1TAB(4)"{RVS}"A$(R,1)"{OFF}"TAB(21)R-17+14
"{RVS}"A$(R+13,1)"{OFF}" :rem 168
1180 NEXT :rem 9
1190 GOSUB380 :rem 230
1200 IFB%>32ORB%<1THEN1110 :rem 195
1210 GOSUB420 :rem 218
1220 RETURN :rem 165
1230 REM:STAMPA SUBMENU PESCE :rem 57
1240 PRINT"{CLR}":I=103 :rem 127
1250 FORU=1TO40:PRINT"Q[<+>]";:NEXT:PRINT :rem 118
1260 PRINTSPC(18)"{RVS}PESCE{GIU'}" :rem 205
1270 FORR=103TO112 :rem 11
1280 PRINTR-103+1TAB(4)"{RVS}"A$(R,1)"{OFF}"TAB(17)R-103+
11" {RVS}"A$(R+10,1)"{OFF}" :rem 1
1290 NEXT :rem 11
1300 GOSUB380 :rem 223
1310 IFB%>20ORB%<1THEN1240 :rem 198
1320 GOSUB420 :rem 220
1330 RETURN :rem 167
1340 REM:STAMPA SUBMENU ORTAGGI-LEGUMI :rem 200
1350 PRINT"{CLR}":I=43 :rem 84
1360 FORT=1TO10:PRINT"SA{ 2 SPAZI}";:NEXT:PRINT :rem 145
1370 PRINTSPC(13)"{GIU'}{RVS}ORTAGGI*LEGUMI{GIU'}"
:rem 101
1380 FORR=43TO56 :rem 183
1390 PRINT"{RVS}"R-43+1TAB(4)"{OFF}"A$(R,1)TAB(20)"{RVS}"
R-43+15" {OFF}"A$(R+14,1) :rem 171
1400 NEXT :rem 4
1410 GOSUB380 :rem 225
1420 IFB%>28ORB%<1THEN1350 :rem 210
1430 GOSUB420 :rem 222
1440 RETURN :rem 169
1450 REM:STAMPA SUBMENU UOVA-FORMAGGI :rem 129
1460 PRINT"{CLR}":I=123 :rem 133
1470 FORL=1TO20:PRINT"E[<*>]";:NEXT:PRINT :rem 128
1480 PRINTSPC(13)"{RVS}UOVA & FORMAGGI{GIU'}" :rem 9
1490 FORR=123TO133 :rem 20
1500 PRINTR-123+1TAB(4)"{RVS}"A$(R,1)"{OFF}"TAB(18)R-123+
12" {RVS}"A$(R+11,1)"{OFF}" :rem 3
1510 NEXT :rem 6
1520 PRINT"{GIU'}{RVS}NB:{OFF}2 UOVA CORRISPONDONO A CIRC
A 100 GR." :rem 111

```

1530	GOSUB380	:rem	228
1540	IFB%>22ORB%<1THEN1460	:rem	209
1550	GOSUB420	:rem	225
1560	RETURN	:rem	172
1570	REM:STAMPA SUBMENU FRUTTA	:rem	166
1580	PRINT"{CLR}":I=71	:rem	90
1590	FORT=1TO26:PRINT"Z Q";:NEXT:PRINT	:rem	180
1600	PRINTSPC(17)"{GIU'}{RVS}FRUTTA{GIU'}"	:rem	65
1610	FORR=71TO82	:rem	179
1620	PRINTR-71+1TAB(4)"{RVS}"A\$(R,1)TAB(19)"{OFF}"R-71+13 " {RVS}"A\$(R+12,1)	:rem	215
1630	NEXT	:rem	9
1640	GOSUB380	:rem	230
1650	IFB%>24ORB%<1THEN1580	:rem	216
1660	GOSUB420	:rem	227
1670	RETURN	:rem	174
1680	REM:STAMPA SUBMENU BEVANDE	:rem	199
1690	PRINT"{CLR}":I=177	:rem	147
1700	FORY=1TO40:PRINT"A ";:NEXT:PRINT	:rem	196
1710	PRINTTAB(16)"{RVS}BEVANDE{GIU'}"	:rem	65
1720	FORR=177TO185	:rem	32
1730	PRINTR-176TAB(4)"{RVS}"A\$(R,1)"{OFF}"TAB(18)R-176+9" {RVS}"A\$(R+9,1)"{OFF}"	:rem	105
1740	NEXT:PRINT	:rem	210
1750	PRINT"{RVS}NB:{OFF}LE BEVANDE ANALCOLICHE PRODUCONO"	:rem	94
1760	PRINT"{ 3 SPAZI}CALORIE SECONDO LA % DI ZUCCHERI"	:rem	69
1770	GOSUB380	:rem	234
1780	IFB%>18ORB%<1THEN1690	:rem	225
1790	GOSUB420	:rem	231
1800	RETURN	:rem	169
1810	REM:STAMPA SUBMENU DOLCI	:rem	56
1820	PRINT"{CLR}":I=195	:rem	142
1830	FORMM=1TO20:PRINT"{RVS}ZQ";:NEXT:PRINT	:rem	3
1840	PRINTTAB(18)"{GIU'}{RVS}DOLCI{ 2 GIU'}"	:rem	223
1850	FORR=195TO205	:rem	29
1860	PRINTR-195+1TAB(4)"{RVS}"A\$(R,1)"{OFF}"TAB(18)R-195+ 12"{RVS}"A\$(R+11,1)	:rem	72
1870	NEXT	:rem	15
1880	GOSUB380	:rem	236
1890	IFB%>22ORB%<1THEN1820	:rem	217
1900	GOSUB420	:rem	224
1910	RETURN	:rem	171
1920	REM:STAMPA SUBMENU CONDIMENTI	:rem	185
1930	PRINT"{CLR}":I=95	:rem	95
1940	FORY=1TO20:PRINT"S A ";:NEXT:PRINT	:rem	155
1950	PRINTTAB(15)"{GIU'}{RVS}CONDIMENTI{ 2 GIU'}"	:rem	93
1960	FORR=95TO102	:rem	234

```

1970 PRINTTAB(10)R-95+1"{ 2 SPAZI}{RVS}"A$(R,1)"{OFF}"
:rem 193
1980 NEXT
:rem 17
1990 GOSUB380
:rem 238
2000 IFB%>8ORB%<1THEN1930
:rem 159
2010 GOSUB420
:rem 217
2020 RETURN
:rem 164
2030 REM:COMPUTO CALORIE TOTALI*VALUTAZIONE DELLA DIETA
:rem 27
2040 PRINT"{CLR}{ 2 GIU' }":FORN=1TO9:PRINT"{ 3 SPAZI}S";
NEXT:PRINT
:rem 94
2050 IFCT=0THEN2100
:rem 76
2060 H=AT/CT*100:REM:CALCOLO %
:rem 221
2070 J=VT/CT*100:REM:DEI PRINCIPI
:rem 3
2080 K=GT/CT*100:REM:NUTRITIVI
:rem 148
2090 L=ZT/CT*100
:rem 131
2100 PRINT"{GIU'} GLI ALIMENTI DA TE SCELTI"
:rem 182
2110 PRINT"{GIU'} CONTENGONO COMPLESSIVAMENTE"
:rem 95
2120 PRINTTAB(25)"{GIU'}CALORIE"TAB(34)"%"
:rem 45
2130 PRINT"{GIU'} {RVS} CALORIE TOTALI"TAB(20)"{ 4 C}> "C
T;TAB(34)"100"
:rem 212
2140 PRINT"{GIU'} {RVS} PROTEINE ANIMALI"TAB(20)"{ 4 C}
> "AT;TAB(34)LEFT$(STR$(H),5)
:rem 74
2150 PRINT"{GIU'} {RVS} PROTEINE VEGETALI"TAB(20)"
{ 4 C}> "VT;TAB(34)LEFT$(STR$(J),5)
:rem 184
2160 PRINT"{GIU'} {RVS} GRASSI"TAB(20)"{ 4 C}> "GT;TAB(34)
LEFT$(STR$(K),5)
:rem 189
2170 PRINT"{GIU'} {RVS} CARBOIDRATI"TAB(20)"{ 4 C}> "ZT;T
AB(34)LEFT$(STR$(L),5)
:rem 45
2180 PRINT"{GIU'} {RVS} VITAMINE"TAB(20)"{ 4 C}>"VT$
:rem 2
2190 PRINT"{GIU'}{ 2 SPAZI}*****"
:rem 150
2200 PRINT"{GIU'}VUOI ALTRI ALIMENTI? (S/N)
:rem 246
2210 GETTU$:IFTU$=""THEN2210
:rem 125
2220 IFTU$="S"THEN5550
:rem 244
2230 IFTU$<>"S"ANDTU$<>"N"THEN2210
:rem 212
2240 RETURN
:rem 168
:rem 0

```


Keypad

Le straordinarie capacità di gestione della memoria offerte dal C64 permettono modifiche, anche radicali, dell'Interprete Basic e del Sistema Operativo (Kernal). È infatti possibile ricopiare la memoria ROM (Basic + S.O.) nella RAM avente gli stessi indirizzi. Come saprete, il contenuto della ROM può venir cambiato a piacimento, in modo ad esempio da personalizzare l'Interprete Basic. Recentemente (Papersoft n. 24) abbiamo infatti pubblicato il programma "User Basic", che consente di alterare le parole-chiave del linguaggio Basic sostituendole con altre parole della stessa lunghezza.

KEYPAD permette invece di ridefinire la tastiera del C64, realizzando un tastierino numerico strutturato come quello delle calcolatrici elettroniche, utilissimo quando si devono digitare dati numerici. Dopo aver ricopiato la ROM in RAM (impiegando poco più di un minuto), vengono memorizzate due routine in linguaggio macchina nell'area RAM che inizia da 49152 (\$C000), senza rubare neanche un byte alla memoria Basic. A questo punto compare la scritta READY e il programma è attivato: premendo F1 avremo a disposizione il tastierino numerico (e il bordo dello schermo diverrà rosso), mentre con F3 torneremo alla tastiera tradizionale

(bordo rosa). Le configurazioni della tastiera nei due casi sono le seguenti:

7	8	9
U	I	O (standard)
J	K	L
M		

7	8	9
4	5	6 (tastierino)
1	2	3
0		

Vi consigliamo di applicare ai tasti interessati delle piccole etichette autoadesive che ricordino il numero ad essi associato. KEYPAD viene disabilitato se si preme RUN-STOP e RESTORE; per riattivarlo è sufficiente digitare SYS 50000. Chi volesse modificare i colori che segnalano il tipo di tastiera in uso deve agire sulle locazioni 49163 (colore bordo con tastiera in RAM) e 49179 (colore bordo con tastiera in ROM).

Il programma può essere facilmente modificato per passare dalla tastiera anglosassone (QWERTY) a quella italiana (QZERTY), che prevede la Z al posto della W e la M al posto dei due punti. È sufficiente cancellare le linee 240 e 260 e togliere la REM all'inizio della linea 250. L'uso dei tasti funzione per commutare tra le due configurazioni è analogo a quello già descritto.

2 REM	K E Y P A D	:rem	212
3 REM	PER C 64	:rem	171
5 REM	BY DAVID SANTINOLI	:rem	205
20 REM	COPIA BASIC + S.O.	:rem	61
40	PRINT"{CLR}{GIU'}{ 11 SPAZI}ATTENDERE PREGO"	:rem	242
50	FORI=40960TO49151:POKEI,PEEK(I):NEXT	:rem	146
60	FORI=57344TO65535:POKEI,PEEK(I):NEXT	:rem	155
80	REM ROUTINES IN L.M.	:rem	81
100	FORI=49152TO49185:READA:POKEI,A:NEXT	:rem	224
110	FORI=50000TO50012:READA:POKEI,A:NEXT	:rem	190
120	SYS50000	:rem	135
140	DATA 165,203,201,004,208,012,169	:rem	220
150	DATA 053,133,001,169,066,141,032	:rem	223
160	DATA 208,076,049,234,201,005,208	:rem	232
170	DATA 009,169,055,133,001,169,090	:rem	238
180	DATA 141,032,208,076,049,234	:rem	39
190	DATA 120,169,192,141,021,003,169	:rem	231
200	DATA 000,141,020,003,088,096	:rem	18
220	REM VALORI PER KEYPAD NUMERICO	:rem	76
240	FORI=48TO54:READA:POKE60300+A,I:NEXT	:rem	206
250	REM POKE60298,90:POKE60301,87:POKE60325,58:POKE60334,		
	77	:rem	75
260	DATA 25,23,26,31,19,22,27	:rem	137

Indovina i colori

C64

Questo è un gioco che mette alla prova la memoria visiva; possono partecipare fino a 9 persone. Lanciando il programma appaiono sul video alcune brevi istruzioni, dopo di che viene richiesto il numero ed il nome dei partecipanti. Inizia quindi il gioco vero e proprio: 18 bandierine colorate in rosa, verde e giallo, disposte a triangolo, vengono mostrate per mezzo minuto. In questo tempo i concorrenti devono cercare di memorizzare la disposizione dei colori. Allo scadere dei 30 secondi le bandierine diventano tutte uguali, e a turno i giocatori devono tentare, partendo dall'angolo inferiore sinistro e proseguendo verso destra ad arrivare al vertice, di ricostruire i colo-

ri. Ad una ad una le bandierine lampeggiano, e il giocatore deve premere il tasto corrispondente all'iniziale del colore (R=rosa, V=verde e G=giallo); se sbaglia passa la mano al concorrente successivo, e via di seguito finché tutti avranno avuto a disposizione tre tentativi. Al termine viene mostrata la classifica finale: il vincitore naturalmente sarà chi avrà indovinato il maggior numero di colori. Chi ritenesse eccessivi i 30 secondi può modificare la stringa "000030" alla linea 520, sostituendola ad esempio con "000020" (per 20 secondi), e cambiare di conseguenza il messaggio stampato alla linea 220 (istruzioni di gioco).

```
0 REM ***** :rem 46
1 REM * "INDOVINA{ 2 SPAZI}I{ 2 SPAZI}COLORI" * :rem 22
5 REM * DI ROBERTO MICHELOTTI * :rem 9
8 REM ***** :rem 54
9 REM DIMENSIONA MATRICI-AZZERA IL SID LEGGE I DAT
  I-PREPARA LA STRINGHE :rem 116
10 PRINT "{CLR}":DIMCR(1,17),A$(4),BA$(17),NG$(10),PG(10):
  POKE53280,6:POKE53281,0 :rem 199
13 FORA=54272TO54296:POKEA,0:NEXT :rem 249
16 POKE54296,15:POKE54276,32:POKE54277,60:POKE54278,128:P
  OKE54272,25 :rem 114
20 FORA=0TO4:READB:A$(A)=CHR$(B):NEXT :rem 225
30 FORC=0TO18:READB:FORA=0TO4 :rem 68
40 A$(A)=A$(A)+CHR$(B):NEXTA,C :rem 119
50 FORB=0TO17:FORA=0TO1:READC :rem 66
60 CR(A,B)=C:NEXTA,B` :rem 163
63 : :rem 163
64 REM FA LA CORNICE DI BANDIERE :rem 44
65 X=0:A=1 :rem 25
70 FORY=1TO36STEP5:GOSUB1800 :rem 21
80 PRINTA$(A):A=A+1:IFA>3THENA=1 :rem 154
```


90	NEXT: IFX=0 THEN X=20: GOTO 70	:rem	54
100	Y=1	:rem	88
110	FORX=5 TO 15 STEP 5: GOSUB 1800	:rem	64
120	PRINTA\$(A): A=A+1: IF A>3 THEN A=1	:rem	197
130	NEXT: IF Y=1 THEN Y=36: GOTO 110	:rem	150
134	:	:rem	210
135	ONSGOTO 810	:rem	91
140	PRINTCHR\$(19)CHR\$(158)	:rem	206
150	PRINTSPC(131)"INDOVINA I COLORI"	:rem	156
160	PRINTTAB(10)"-----"	:rem	72
170	PRINTTAB(5)"IN QUESTO GIOCO OCCORRE BUONA"	:rem	49
180	PRINTTAB(5)"MEMORIA E COLPO D'OCCHIO."	:rem	225
190	PRINTTAB(5)"18 BANDIERINE, DI 3 COLORI DI-"	:rem	112
200	PRINTTAB(5)"VERSI, APPAIONO ALL'INIZIO DEL"	:rem	115
210	PRINTTAB(5)"GIOCO DISPOSTE A TRIANGOLO; SI"	:rem	100
220	PRINTTAB(5)"HANNO 30 SECONDI DI TEMPO PER"	:rem	151
230	PRINTTAB(5)"MEMORIZZARE LA SEQUENZA DEI"	:rem	227
240	PRINTTAB(5)"COLORI PARTENDO DAL BASSO E"	:rem	119
250	PRINTTAB(5)"DA SINISTRA VERSO DESTRA FINO"	:rem	53
260	PRINTTAB(5)"AD ARRIVARE AL VERTICE."	:rem	116
270	PRINTTAB(5)"TRASCORSO IL TEMPO LE BANDIE-"	:rem	2
280	PRINTTAB(5)"RINE DIVENTANO TUTTE UGUALI."	:rem	41
290	PRINTTAB(12)CHR\$(5)"PREMI LO SPAZIO"CHR\$(158)	:rem	92
300	IFPEEK(203)<>60 THEN 300	:rem	207
310	PRINTCHR\$(19)SPC(120);	:rem	226
320	FORA=0 TO 15	:rem	51
330	PRINTTAB(5)"{ 30 SPAZI}":REM 30 SPAZI	:rem	204
340	NEXT	:rem	214
350	PRINTCHR\$(19)CHR\$(158)SPC(125)"LO SCOPO DEL GIOCO E' RICOR-{ 2 SPAZI}"	:rem	97
360	PRINTTAB(5)"DARSI LA SEQUENZA DEI COLORI. "	:rem	251
370	PRINTTAB(5)"SI PUO' GIOCARE DA 1 A 9 PER- "	:rem	189
380	PRINTTAB(5)"SONE, ALTERNANDOSI QUANDO UNO "	:rem	104
390	PRINTTAB(5)"SBAGLIA. SI HANNO 3 TENTATIVI "	:rem	230
400	PRINTTAB(5)"A TESTA: VINCE CHI SI RICORDA "	:rem	167
410	PRINTTAB(5)"PIU' COLORI.{ 18 SPAZI}":PRINTCHR\$(5)	:rem	28
420	PRINTTAB(5)"NUMERO DEI GIOCATORI?{ 7 SPAZI}";:POKE198,0	:rem	238
430	GETZ\$: IFZ\$<"1"ORZ\$>"9" THEN 430	:rem	141
440	PRINTZ\$CHR\$(17)CHR\$(17)CHR\$(17)	:rem	209
450	FORA=1 TO VAL(Z\$)	:rem	132
460	PRINTCHR\$(145)TAB(5)"NOME GIOC."A;:INPUTNG\$(A)	:rem	196
470	PRINTCHR\$(145)TAB(5)"{ 30 SPAZI}":REM 30 SPAZI	:rem	189
480	NG\$(A)=LEFT\$(NG\$(A),10):NEXT	:rem	21
490	PRINTCHR\$(17)TAB(4)"PREMI SPAZIO E OCCHIO AI COLORI!"	:rem	68

500 IFPEEK(203)<>60THEN500	:rem	211
510 GOSUB1500:TI\$="000000"	:rem	120
520 IFTI\$<"000030"THEN520	:rem	80
528 :	:rem	217
529 REM INIZIO GIOCO	:rem	199
530 FORA2=1TO3	:rem	54
540 FORA1=1TOVAL(Z\$):B%=0:GOSUB1600:A=0	:rem	44
550 X=22:Y=0:GOSUB1800	:rem	22
560 PRINTCHR\$(5)NG\$(A1)", DI CHE COLORE ERA? ":POKE198,0	:rem	58
570 IFA=18THEN680	:rem	223
580 GETR\$:IFR\$=""THENGOSUB1900:GOTO580	:rem	60
590 IFR\$<>"R"ANDR\$<>"G"ANDR\$<>"V"THEN580	:rem	42
600 IFR\$="R"THENR\$=CHR\$(150)+"ROSA{ 2 SPAZI}":GOTO625	:rem	236
610 IFR\$="G"THENR\$=CHR\$(158)+"GIALLO":GOTO625	:rem	109
620 R\$=CHR\$(30)+"VERDE "	:rem	229
625 X=22:Y=30:GOSUB1800:PRINTR\$:rem	137
630 IFLEFT\$(R\$,1)=LEFT\$(BA\$(A),1)THENJ=1:PG=PG+1	:rem	246
640 X=CR(0,A):Y=CR(1,A):GOSUB1800	:rem	137
650 PRINTBA\$(A)	:rem	97
660 IFJTHENA=A+1:J=0:GOSUB2000:GOTO570	:rem	59
670 GOSUB2100:FORRI=1TO2000:NEXT:GOSUB1700	:rem	106
680 IFA>PG(A1)THENPG(A1)=A	:rem	13
690 NEXTA1,A2:A3=0	:rem	10
698 :	:rem	225
699 REM ORDINA I GIOC. IN BASE AI PUNTI	:rem	174
700 A2=A3+1:A1=A3+2:A3=A3+1	:rem	140
710 IFA3>VAL(Z\$)THEN800	:rem	82
720 FORA=A1TOVAL(Z\$)	:rem	197
730 IFPG(A2)>PG(A)THENNEXT	:rem	89
740 IFA>VAL(Z\$)THEN770	:rem	40
750 A2=A:A1=A+1	:rem	19
760 IFA1<=VAL(Z\$)THEN720	:rem	145
770 PG=PG(A2):NG\$=NG\$(A2)	:rem	122
780 PG(A2)=PG(A3):NG\$(A2)=NG\$(A3)	:rem	5
790 PG(A3)=PG:NG\$(A3)=NG\$:GOTO700	:rem	136
798 :	:rem	226
799 REM STAMPA LA CLASSIFICA E ASSOCIA	I GIOCATO	
RI A PARI MERITO	:rem	33
800 S=1:PRINTCHR\$(147):GOTO65	:rem	236
810 PRINTCHR\$(19)CHR\$(158)SPC(175)"CLASSIFICA"	:rem	188
820 PRINTTAB(14)"-----":PRINT	:rem	219
830 A=1:K=1	:rem	61
840 IFPG(A+1)=PG(A)THENJ=0:GOTO860	:rem	13
850 J=1	:rem	85
860 IFKTHENA1=A	:rem	151
870 PRINTTAB(4)A1CHR\$(157)")"NG\$(A)TAB(18)"CON"PG(A)TAB(2		
5)"BANDIERINE	:rem	1
880 IFJ=0THENK=0:GOTO900	:rem	217

890 K=1	:rem	90
900 A=A+1:IFA=VAL(Z\$)THEN860	:rem	122
910 IFA<VAL(Z\$)THEN840	:rem	35
919 :	:rem	221
920 PRINT:PRINTCHR\$(5)TAB(10)"GIOCATE ANCORA? (S/N) ":POK		
E198,0	:rem	186
930 GETR\$:IFR\$<>"S"ANDR\$<>"N"THEN930	:rem	98
940 IFR\$="S"THENRUN	:rem	154
950 FORA=54272TO54296:POKEA,0:NEXT	:rem	51
960 PRINT"{CLR}":END	:rem	20
998 :	:rem	228
999 REM LINEE DATA	:rem	22
1000 DATA154,150,158,30,5,18,213,192	:rem	16
1001 DATA201,17,157,157,157,202,192,203	:rem	167
1002 DATA17,157,154,146,182,17,157,184	:rem	136
1003 DATA16,6,16,10,16,14,16,18,16,22	:rem	54
1004 DATA16,26,16,30,12,10,12,14,12,18	:rem	92
1005 DATA12,22,12,26,8,14,8,18,8,22	:rem	216
1006 DATA4,16,4,20,0,18	:rem	135
1498 :	:rem	16
1499 REM STAMPA BANDIERE COLORI RANDOM	:rem	68
1500 PRINTCHR\$(147):FORA=0TO17:A%=1	:rem	41
1510 B=INT(RND(1)*3)+1:IFB=C%THEN1510	:rem	71
1520 POKE214,CR(0,A):POKE211,CR(1,A)	:rem	234
1530 SYS58732:PRINTA\$(B)	:rem	144
1540 BA\$(A)=A\$(B):C%=B	:rem	89
1550 IFA%THENNEXT:A%=0	:rem	59
1560 RETURN	:rem	172
1598 :	:rem	17
1599 REM STAMPA BANDIERE COLORI UGUALI	:rem	75
1600 FORA=0TO17:A%=1	:rem	117
1610 POKE214,CR(0,A):POKE211,CR(1,A)	:rem	234
1620 SYS58732	:rem	209
1630 PRINTA\$(B%):IFA%THENNEXT:A%=0	:rem	30
1640 RETURN	:rem	171
1698 :	:rem	18
1699 REM CANCELLA LINEA DOMANDE	:rem	81
1700 FORC=1904TO1944:POKEC,32:NEXT	:rem	40
1710 RETURN	:rem	169
1711 :	:rem	4
1798 :	:rem	19
1799 REM SOUBROUTINE DI TABULAZIONE	:rem	232
1800 POKE214,X:POKE211,Y	:rem	149
1810 SYS58732:RETURN	:rem	236
1898 :	:rem	20
1899 REM LAMPEGGIO BANDIERE	:rem	142
1900 IFB%=0THENB%=4:GOTO1920	:rem	115
1910 B%=0	:rem	159
1920 GOSUB1610:FORRI=1TO150:NEXT:RETURN	:rem	10

1998 :	:rem	21
1999 REM SUONO COLORE ESATTO	:rem	232
2000 POKE54273,30:POKE54276,33	:rem	136
2010 FORM=1TO150:NEXT	:rem	23
2020 POKE54276,32:RETURN	:rem	166
2098 :	:rem	13
2099 REM SUONO COLORE ERRATO	:rem	221
2100 POKE54273,5:FORM=1TO15	:rem	61
2110 POKE54276,33:FORM1=1TO15:NEXT	:rem	28
2120 POKE54276,32:FORM1=1TO15:NEXTM1,M	:rem	19
2130 RETURN	:rem	166



Super Drums

Se ancora non siete convinti delle possibilità sonore del SID, il chip "musicale" del C64, questo è un programma che farà ricredere anche i più scettici. Questa volta il vostro computer si trasformerà in una batteria elettronica.

Si possono ottenere le varie per-

cussioni usando i tasti F, G, B, J e la barra spaziatrice. Con il tasto R è possibile avere un cocktail casuale dei vari suoni, che può essere interrotto premendo S.

Con Super Drums ognuno può diventare un maestro della batteria elettronica!

1000 REM SUPER DRUMS	:rem	191
1010 PRINTCHR\$(147)	:rem	61
1020 POKE53280,0:POKE53281,0	:rem	24
1030 PRINTTAB(12)"{CYN}{RVS}{ 2 SPAZI}SUPER{ 2 SPAZI}DRUM S{ 2 SPAZI}{ 3 GIU'}"	:rem	30
1090 PRINTTAB(10)"[B] = GRANCASSA"	:rem	238
1100 PRINTTAB(10)"[J] = RULLANTE"	:rem	194
1110 PRINTTAB(10)"[F] = TOM ALTO"	:rem	120
1120 PRINTTAB(10)"[G] = TOM BASSO"	:rem	194
1130 PRINTTAB(10)"[] = PIATTO"	:rem	229
1140 PRINTTAB(10)"[R] = ASSOLO CASUALE"	:rem	54
1150 PRINTTAB(10)"[S] = FINE ASSOLO"	:rem	92
1160 REM INIZIALIZZAZIONE	:rem	127
1170 S=54272:S1=54286:S2=54279	:rem	104
1180 FORI=0TO24:POKEI+S,0:NEXT	:rem	117
1190 POKES+24,15	:rem	112
1200 REM INPUT	:rem	55

1210	GETA\$:rem	9
1220	IFA\$="R"THEN1720	:rem	133
1230	IFA\$="J"THENGOSUB1300	:rem	248
1240	IFA\$="B"THENGOSUB1380	:rem	249
1250	IFA\$="F"THENGOSUB1470	:rem	254
1260	IFA\$="G"THENGOSUB1560	:rem	0
1270	IFA\$=" "THENGOSUB1650	:rem	186
1280	GOTO1210	:rem	200
1290	REM RULLANTE	:rem	23
1300	POKES2+4,0	:rem	51
1310	POKES2+5,8	:rem	61
1320	POKES2+6,0	:rem	55
1330	POKES2+4,129	:rem	162
1340	POKES2,200	:rem	58
1350	POKES2+1,90	:rem	110
1360	RETURN	:rem	170
1370	REM GRANCASSA	:rem	66
1380	POKES1+4,0	:rem	58
1390	POKES2+4,0	:rem	60
1400	POKES1+5,4	:rem	56
1410	POKES1+6,2	:rem	56
1420	POKES1+4,17	:rem	109
1430	POKES1,2	:rem	217
1440	POKES1+1,7	:rem	59
1450	RETURN	:rem	170
1460	REM TOM ALTO	:rem	207
1470	POKES1+4,0	:rem	58
1480	POKES2+4,0	:rem	60
1490	POKES1+5,4	:rem	65
1500	POKES1+6,2	:rem	56
1510	POKES1+4,17	:rem	109
1520	POKES1,2	:rem	217
1530	POKES1+1,20	:rem	102
1540	RETURN	:rem	170
1550	REM TOM BASSO	:rem	23
1560	POKES1+4,0	:rem	58
1570	POKES2+4,0	:rem	60
1580	POKES1+5,4	:rem	65
1590	POKES1+6,2	:rem	65
1600	POKES1+4,17	:rem	109
1610	POKES1,2	:rem	217
1620	POKES1+1,15	:rem	106
1630	RETURN	:rem	170
1640	REM PIATTO	:rem	128
1650	POKES+4,0	:rem	9
1660	POKES+5,9	:rem	20
1670	POKES+6,0	:rem	13
1680	POKES+4,129	:rem	120
1690	POKES,2	:rem	176
1700	POKES+1,150	:rem	104

1710 RETURN	:rem 169
1720 REM ASSOLO	:rem 127
1730 R=INT(RND(TI)*5+1)	:rem 35
1740 D=INT(RND(1)*100)	:rem 170
1750 FORT=1TO50+D:NEXT	:rem 102
1760 ONRGOSUB1300,1380,1470,1560,1650	:rem 225
1770 GETB\$:IFB\$<>"S"THEN1730	:rem 79
1780 GOTO1210	:rem 205

E' IN EDICOLA

Bit,
la prima rivista europea
di personal computer,
software, accessori,
la più prestigiosa
e più diffusa in Italia

con tutta la competenza del



**GRUPPO
EDITORIALE
JACKSON**

SCALDA IL JOYSTICK E GASATI CON



**SONO
IN EDICOLA**



**GRUPPO
EDITORIALE
JACKSON**

Milano-Londra-Madrid-San Francisco

IL VERO GIOCO COMINCIA ADESSO